
SPICE: SIMULACIÓN A NIVEL ELÉCTRICO

- **PRIMITIVAS DE SPICE**
- **SINTAXIS DE SPICE**
- **TIPOS DE ANÁLISIS:**
 - Análisis en DC
 - Análisis en AC
 - Análisis transitorio
 - Análisis a distintas temperaturas
- **ELEMENTOS:**
 - Resistencias
 - Condensadores
 - Bobinas
 - Bobinas acopladas
 - Fuentes independientes
 - Fuentes variables con el tiempo:
 - Pulso
 - Sinusoidal
 - Exponencial
 - Lineal a tramos
 - Modulada en frecuencia
 - Fuentes controladas
- **DISPOSITIVOS SEMICONDUCTORES:**
 - Diodo de unión $p-n$
 - Transistor BJT
 - Transistor JFET
 - Transistor MOSFET
- **SUBCIRCUITOS**
- **LÍNEAS DE CONTROL:**
 - Análisis en DC:
.OP .DC .TF .SENS
 - Análisis en AC:
.AC .NOISE .DISTO
 - Análisis transitorio:
.IC .TRAN .FOUR
 - Control de Salida
- **PROBLEMAS EN LA SIMULACIÓN**
- **ANEXO**

PRIMITIVAS DE SPICE

SPICE es un simulador eléctrico que reproduce el comportamiento (I, V) de circuitos formados por los siguientes elementos o primitivas:

1. Resistencias

2. Condensadores (lineales o no (polinómicos))

3. Inductores (lineales o no)

4. Fuentes independientes de tensión e intensidad

5. Cuatro tipos de fuentes dependientes (lineales o no)


- VCVS (de tensión controlada por tensión)
- VCCS (de intensidad controlada por tensión)
- CCVS (de tensión controlada por intensidad)
- CCCS (de intensidad controlada por intensidad)

6. Dispositivos semiconductores

- Diodos
- BJT *NPN*
- BJT *PNP*
- JFET canal *P*
- JFET canal *N*
- MOSFET canal *P*
- MOSFET canal *N*

7. Líneas de transmisión

EJEMPLO:

FICHERO DE ENTRADA DE SPICE

Ejemplo de circuito. Esta es la línea de TÍTULO.

*

* **Esta línea es un COMENTARIO.**

*

V1 1 0 5V

R1 1 2 10K

R2 2 0 10k

*

*

.OP

.END

FICHERO DE SALIDA DE SPICE

**** 10/16/93 13:15:46 ***** Evaluation PSpice (January 1991) *****
Ejemplo de circuito. Esta es la linea de TITULO.

**** CIRCUIT DESCRIPTION

*
* Esta linea es un COMENTARIO.

*
V1 1 0 5V
R1 1 2 10K
R2 2 0 10k

*
*
.OP
.END

** SMALL SIGNAL BIAS SOLUTION TEMPERATURE =27.000 DEG C

| NODE | VOLTAGE | NODE | VOLTAGE | NODE | VOLTAGE | NODE | VOLTAGE |
|------|---------|------|---------|------|---------|------|---------|
| (1) | 5.0000 | (2) | 2.5000 | | | | |

VOLTAGE SOURCE CURRENTS
NAME CURRENT

V1 -2.500E-04

TOTAL POWER DISSIPATION 1.25E-03 WATTS

** OPERATING POINT INFORMATION TEMPERATURE = 27.000 DEG C

JOB CONCLUDED
TOTAL JOB TIME 1.59

SINTAXIS DE SPICE – FICHERO DE ENTRADA

Es un fichero tipo texto constituido por un conjunto de líneas, cada una de las cuales constituye un ente individual. Existen distintos tipos de líneas:



■ Línea de título y línea de final:

- Son *obligatorias*.
- Acotan el conjunto de líneas que constituyen la descripción del circuito y de la simulación.

■ Líneas de elementos:

- Existe una por elemento.
- Definen la topología de interconexión del elemento y el valor del mismo.

■ Líneas de modelo:

- Existe una por cada modelo distinto que se considere.
- Especifican los parámetros del modelo en cuestión.

■ Líneas de comentarios:

- Son *transparentes* para SPICE.
- Se inicia con *.

■ Líneas de control:

- Permiten especificar el tipo de análisis a realizar y las condiciones del mismo.

SINTAXIS DE SPICE – LÍNEAS DEL FICHERO

Cada línea del fichero de entrada contiene un conjunto de campos que pueden ser de dos tipos:

■ Nombres:

- Deben comenzar por una letra y no pueden contener espacios, comas, “+”, “=”, “;”.

■ Números:

- Enteros o de coma flotante.
- Se permiten exponentes y factores de escala:

```
T = 1E12   G = 1E9   MEG = 1E6   K = 1E3
M = 1E-3   U = 1E-6   N = 1E-9   P = 1E-12   F = 1E-15
```

- Se ignoran:

- 1) Las letras que haya detrás de un número y no sean factores de escala.
- 2) Las letras que sigan a un factor de escala.

Ejemplo: 1000V = 1E3 = 1KV = 1000.0 VOLT

- *Continuación de una línea:* Añadiendo el símbolo “+” al principio del siguiente renglón.

```
Ejemplo de circuito. Esta es la línea
+ de TITULO.
* Esta línea es un
+ COMENTARIO.
V1 1 0 5V
R1 1 2 10K
R2 2 0 10k
.OP
.END
```

SINTAXIS DE SPICE – LÍNEA GENÉRICA DE ELEMENTO

NOMBRE Nudos_de_conexión_de_terminales VALOR <opciones>

La primera letra del nombre es:

R (resistencias)

C (condensadores)

L (bobinas)

K (bobinas acopladas)

T (líneas de transmisión)

V (fuentes independientes de tensión)

I (fuentes independientes de intensidad)

G (VCCS)

E (VCVS)

F (CCCS)

H (CCVS)

D (diodos)

Q (BJTs)

J (JFETs)

M (MOSFETs)

X (subcircuitos)

- Para especificar los dispositivos semiconductores se necesita además una línea de modelo.

TIPOS DE ANÁLISIS

ANÁLISIS EN DC

- *Cálculo del punto de trabajo o punto de operación:* Considera Ls nulas (cortos), Cs nulas (abiertos) y excitaciones estáticas. (por defecto)
- *Evaluación de características estáticas:* Cálculo del punto de trabajo para un rango de valores de la excitación (**.DC**)
- *Información de los modelos en pequeña señal de los dispositivos en el punto de trabajo (**.OP**)*
- *Análisis de sensibilidad en pequeña señal (**.SENS**)*
- *Característica de transferencia, resistencia de salida y resistencia de entrada en pequeña señal (**.TF**)*

ANÁLISIS EN AC

- *Respuesta frecuencial en pequeña señal:* Circuito linealizado alrededor del punto de trabajo y considerando entrada sinusoidal (**.AC**)
- *Análisis de ruido:* Las fuentes de ruido se calculan automáticamente (**.NOISE**)
- *Análisis de distorsión:* Se superponen en la entrada una o varias señales de distintas frecuencias (**.DISTO**)

ANÁLISIS EN TRANSITORIO

- *Análisis temporal de variables de salida*: Se pueden especificar distintas excitaciones: pulsos, exponenciales, sinusoidales, etc. (**.TRAN**)
- *Análisis de Fourier*: Distintas componentes de Fourier de la salida para una entrada sinusoidal (**.FOUR**)

ANÁLISIS A DIFERENTES TEMPERATURAS

- Las resistencias y algunos parámetros de los dispositivos semiconductores varían con la temperatura (**.TEMP**) (por defecto es 27°C)


SPICE calcula el punto de trabajo del circuito antes que cualquier otro tipo de análisis para:

- ⇒ Linealizar modelos. (análisis en AC)
- ⇒ Condiciones iniciales ($t = 0$). (análisis transitorio)

ELEMENTOS PASIVOS

RESISTORES:

Rxxxxxxx N+ N- VALOR <TC=TC1 <TC2>>




TC1 y TC2 son coeficientes opcionales de temperatura (nulos por defecto)


$$\text{VALOR (T)} = \text{VALOR (T}_0) \left[1 + T_{C1}(T-T_0) + T_{C2}(T-T_0)^2 \right]$$

CONDENSADORES Y BOBINAS:

Cxxxxxxx N+ N- VALOR <IC=val>
Lxxxxxxx N+ N- VALOR <IC=val>





$$IC = V_{Cxxxxxxx}(t=0) \text{ (V)}$$



$$IC = I_{Lxxxxxxx}(t=0) \text{ (A)}$$

CONDENSADORES Y BOBINAS NO-LINEALES:


| | | | | | |
|------------------|----|----|------|-----------|----------|
| C xxxxxxx | N+ | N- | POLY | C0 ... CM | <IC=val> |
| L xxxxxxx | N+ | N- | POLY | L0 ... LM | <IC=val> |



BOBINAS ACOPLADAS:

| | | | |
|------------------|----------|----------|-------|
| K xxxxxxx | Lyyyyyyy | Lzzzzzzz | VALOR |
|------------------|----------|----------|-------|


teniendo en cuenta que $0 < \text{VALOR} < 1$



FUENTES INDEPENDIENTES


Se pueden dar dos situaciones básicas de especificaciones:

■ Caso A:



- En el análisis en DC o en transitorio se anula la fuente de AC.
- En el análisis en AC éste se realiza sobre el punto de trabajo determinado por el valor de la fuente de DC.

■ Caso B:



- Para análisis transitorio la fuente es variable en el tiempo y se anula la de AC.
- El análisis en DC se hace con $f(0)$.
- El análisis en AC, éste se hace sobre el punto de trabajo resultante con $f(0)$.

Tanto en el caso A como en el B el formato es el mismo:

Vxxxxxxx N+ N- <<DC> Valor DC/TRAN> <AC <ACMAG <ACPH>>>

Ixxxxxxx N+ N- <<DC> Valor DC/TRAN> <AC <ACMAG <ACPH>>>

- DC y AC delimitan qué parte de la información corresponde a la fuente DC/TRAN y cuál a la fuente de AC.
- DC es opcional, normalmente sólo se usa cuando se va a hacer un barrido del valor de la fuente (.DC).

■ **VALORES POR DEFECTO:**

Valor DC/TRAN = 0
AC = no hay fuente de AC

- Si hay fuente de AC:

ACMAG = 1
ACPHASE = 0°

FUENTES VARIABLES EN EL TIEMPO


Hay 5 tipos de fuentes variables en el tiempo:

- ➔ PULSO (PULSE)
- ➔ SINUSOIDAL (SIN)
- ➔ EXPONENCIAL (EXP)
- ➔ LINEAL A TRAMOS (PWL, Piece Wise Linear)
- ➔ MODULADA EN FRECUENCIA (SFFM, Single Frequency FM)

FUENTES VARIABLES EN EL TIEMPO

PULSO:

PULSE (V1 V2 TD TR TF PW PERIODO)



- Esta forma de onda se repite cada $\Delta t = \text{PERIODO}$.
- Si algún parámetro se omite o se pone a cero SPICE asume los siguientes **VALORES POR DEFECTO**:


| PARÁMETRO | VALOR DEF. | UNIDAD |
|------------------------|-------------|--------|
| V1 (valor inicial) | obligatorio | V o A |
| V2 (valor final) | obligatorio | V o A |
| TD (tiempo de retraso) | 0.0 | s |
| TR (tiempo de subida) | TSTEP | s |
| TF (tiempo de bajada) | TSTEP | s |
| PW (tiempo a V2) | TSTOP | s |
| PERIODO | TSTOP | s |

donde TSTEP y TSTOP son parámetros del análisis transitorio (Ver formato de la línea de control .TRAN).

FUENTES VARIABLES EN EL TIEMPO

EJEMPLO

PULSE (0 5 10NS 5NS 5NS 20NS 50NS)



o bien:

PULSE (0V 5V 10N 5N 5N 20N 50N)


PULSE (0V 5V 10NS 5NS 5NS 20NS 50NS)

pulse (0 5 10n 5n 5n 20n 50n)

FUENTES VARIABLES EN EL TIEMPO

SINUSOIDAL:

`SIN (V0 VA FREQ <TD> <THETA>)`



Se describe por:

$$\begin{aligned}
 &V_0 \quad \text{si } 0 < t < T_D \\
 &V_0 + V_A \cdot e^{-(t-T_D) \cdot \text{THETA}} \cdot \text{sen}[2\pi \cdot \text{FREQ} \cdot (t-T_D)] \quad \text{si } t \geq T_D
 \end{aligned}$$


■ VALORES POR DEFECTO:

| PARÁMETRO | VALOR DEF. | UNIDAD |
|-----------------------------------|-------------|--------|
| V0 (offset) | obligatorio | V o A |
| VA (amplitud) | obligatorio | V o A |
| FREQ (frecuencia) | 1/TSTOP | Hz |
| TD (retraso) | 0.0 | s |
| THETA (factor de amortiguamiento) | 0.0 | 1/s |

FUENTES VARIABLES EN EL TIEMPO

EXPONENCIAL:

EXP (V1 V2 TD1 TAU1 TD2 TAU2)



Se describe por:

$$V_1 \quad \text{si } 0 < t < TD_1$$

$$V_1 + (V_2 - V_1) \cdot (1 - e^{-(t - TD_1)/TAU_1}) \quad \text{si } TD_1 < t < TD_2$$

$$V_1 + (V_2 - V_1) \cdot (1 - e^{-(t - TD_1)/TAU_1}) + (V_1 - V_2) \cdot (1 - e^{-(t - TD_2)/TAU_2}) \quad \text{si } t \geq TD_2$$


■ VALORES POR DEFECTO:

| PARÁMETRO | VALOR DEF. | UNIDAD |
|---------------------------------|-------------|--------|
| V1 (valor inicial) | obligatorio | V o A |
| V2 (valor del pulso) | obligatorio | V o A |
| TD1 (retraso de subida) | 0.0 | s |
| TAU1 (cte. de tiempo de subida) | TSTEP | s |
| TD2 (retraso de bajada) | TD1+TSTEP | s |
| TAU2 (cte. de tiempo de bajada) | TSTEP | s |

FUENTES VARIABLES EN EL TIEMPO

LINEAL A TRAMOS:

PWL (T0 V0 <T1 V1 <T2 V2 <T3 V3 ...>)





■ VALORES POR DEFECTO:

| PARÁMETRO | VALOR DEF. | UNIDAD |
|---------------------------|------------|--------|
| V (valor) | 0.0 | V o A |
| TD1 (retraso del valor V) | TSTEP | s |

EJEMPLO


- ① PWL (0 0 10n 0 15n 5 35n 5 40n 0)
 ② PWL (0 0 10n 0 35n 5 40n 0)



FUENTES VARIABLES EN EL TIEMPO

MODULADA EN FRECUENCIA:

SFFM (V0 VA FC MDI FS)



Se describe por:

$$V = V0 + VA \cdot \text{sen}[2\pi \cdot FC \cdot t + MDI \cdot \text{sen}(2\pi \cdot FS \cdot t)]$$



■ VALORES POR DEFECTO:

| PARÁMETRO | VALOR DEF. | UNIDAD |
|------------------------------|-------------|--------|
| V0 (offset) | obligatorio | V o A |
| VA (amplitud) | obligatorio | V o A |
| FC (frecuencia de portadora) | 1/TSTOP | Hz |
| MDI (índice de modulación) | | |
| FS (frecuencia de señal) | 1/TSTOP | Hz |

FUENTES CONTROLADAS LINEALES

VCVS (Voltage Controlled Voltage Source):



EXXXXXXXX N+ N- NC+ NC- VALOR



$$\text{VALOR} = \frac{V_2}{V_1}$$

VCCS (Voltage Controlled Current Source):



GXXXXXXXX N+ N- NC+ NC- VALOR



$$\text{VALOR} = \frac{I_2}{V_1} \text{ } \bar{\text{U}}$$

CCVS (Current Controlled Voltage Source):



HXXXXXXXX N+ N- VNAM VALOR



$$\text{VALOR} = \frac{V_2}{I_1} \text{ } \Omega$$

CCCS (Current Controlled Current Source):

FXXXXXXXX N+ N- VNAM VALOR



$$\text{VALOR} = \frac{I_2}{I_1}$$

FUENTES CONTROLADAS NO-LINEALES

Tienen varias variables de control y se modelan mediante funciones polinómicas.

■ FORMA GENÉRICA:

$$y = f(x_1, x_2, x_3, \dots, x_m)$$

donde

y, variable controlada, puede ser una tensión o una intensidad.

x_i ($1 \leq i \leq m$), variables de control, son tensiones o intensidades.

Asignación de coeficientes:

$$\begin{aligned} y = & P_0 + P_1 \cdot x_1 + P_2 \cdot x_2 + \dots + P_m \cdot x_m + \\ & + P_{m+1} \cdot x_1^2 + P_{m+2} \cdot x_1 \cdot x_2 + \dots + P_{m+m} \cdot x_1 \cdot x_m + \\ & + P_{2m+1} \cdot x_2^2 + P_{2m+2} \cdot x_2 \cdot x_3 + \dots \end{aligned}$$

■ FORMATO:

VCVS: Exxxxxxx N+ N- <POLY(ND)> NC1+ NC1- <NC2+ NC2- ...>
+ P0 <P1 ...> <IC=val>

donde ND indica el número de variables de control y (NC1+, NC1-), (NC2+, NC2-), ..., son pares de nudos entre los que están definidas dichas variables.

VCCS: El mismo formato anterior cambiando la E inicial por G.

CCVS: Hxxxxxxx N+ N- <POLY(ND)> VN1 <VN2 ...>
+ P0 <P1 ...> <IC=val>

donde VN1, VN2, ..., son nombres de fuentes de tensión donde se miden las intensidades de control.

CCCS: El mismo formato anterior cambiando la H inicial por F.

<POLY(ND)> = por defecto dimensión 1.

<P1 ...> = por defecto nulos. Si sólo se especifica un coeficiente y la dimensión es 1, el valor se asigna a P1.

<IC = ...> = por defecto nulos.


DISPOSITIVOS SEMICONDUCTORES

- Su descripción requiere dos líneas:
 - Una **línea de elemento** donde se especifica la topología y algunos parámetros.
 - Una **línea de control** donde se especifica el **modelo**. Varios dispositivos semiconductores del mismo tipo pueden compartir el mismo modelo.

LÍNEA DE ELEMENTO

DIODO DE UNIÓN:

| |
|--|
| <code>Dxxxxxxx N+ N- MNAME <AREA> <OFF> <IC=V(0)> <M=val></code> |
|--|



VALORES POR DEFECTO:

AREA = Factor multiplicativo de área (escala el valor de corrientes, resistencias y capacidades) (AREA=1).

Dispositivo en ON para condiciones iniciales, análisis DC.



V(0) = 0

M = Multiplicidad (n° de dispositivos iguales conectados en paralelo) (M=1).

- Se pueden especificar dos tipos distintos de condiciones iniciales:
 - Mediante la opción OFF, para mejorar la convergencia en DC.
 - Condiciones iniciales IC, para el análisis transitorio.

TRANSISTOR BIPOLAR:

Qxxxxxxx NC NB NE <NS> MNAME <AREA> <OFF>
 + <IC=VBE,VCE> <M=val>



VALORES POR DEFECTO:

AREA = 1



Dispositivo en ON para condiciones iniciales, análisis DC.

VBE y/o VCE = 0

M = 1

TRANSISTOR JFET:

Jxxxxxxx ND NG NS MNAME <AREA> <OFF>
 + <IC=VDS,VGS> <M=val>



VALORES POR DEFECTO:

AREA = 1


Dispositivo en ON para condiciones iniciales, análisis DC.

VDS y/o VGS = 0

M = 1

TRANSISTOR MOSFET:

Mxxxxxxx ND NG NS NB MNAME <L=val> <W=val>
 + <AD=val> <AS=val> <PD=val> <PS=val> <NRD=val>
 + <NRS=val> <OFF> <IC=VDS, VGS, VBS> <M=val>



VALORES POR DEFECTO:

- L, W = Longitud y anchura del canal.
 Por defecto se especifican mediante tarjeta de control .OPTIONS. (L=100 μ m, W=100 μ m)
- AD, AS = Areas de las difusiones de fuente y drenador en m².
 También especificables mediante línea de control .OPTIONS. (AD=0, AS=0)
- PD, PS = Perímetro de las uniones de drenador y fuente. (PD=0, PS=0)
- NRD, NRS = Número equivalente de cuadrados de las difusiones de drenador y fuente. (NRD=1, NRS=1)
- Dispositivo en ON para condiciones iniciales, análisis DC.
- VDS, VGS y/o VBS = 0
- M = 1

LÍNEA DE MODELO

Se usa para especificar los modelos de los dispositivos semiconductores

.MODEL MNAME TIPO <PAR1=val, PAR2=val ...>

MNAME: Nombre del modelo

TIPO: Especifica el dispositivo semiconductor

D → DIODO
NPN → BJT
PNP → BJT
NJF → JFET
PJF → JFET
NMOS → MOSFET
PMOS → MOSFET

PAR1, PAR2 ...: Parámetros del modelo cuyo valor se va a especificar.
Por defecto SPICE asigna un conjunto de valores.

PARÁMETROS DEL MODELO DEL DIODO

| | NOMBRE | PARÁMETRO | UNID. | VAL. DEF. | EJEMP. |
|----|--------|--|-------|-----------|---------|
| 1 | IS | Saturation current | A | 1.0E -14 | 1.0E-14 |
| 2 | RS | Parasitic resistance | Ω | 0.0 | 10 |
| 3 | N | Emission coefficient | | 1.0 | |
| 4 | TT | Transit time | s | 0.0 | 0.1ns |
| 5 | CJ0 | Zero-bias <i>pn</i> junction capacitance | F | 0.0 | 2pF |
| 6 | VJ | Junction potential | V | 1.0 | 0.6 |
| 7 | M | Junction grading coefficient | | 0.5 | 0.5 |
| 8 | EG | Activation energy | eV | 1.11 | 11.1 |
| 9 | XTI | IS temperature exponent | | 3 | 3 |
| 10 | KF | Flicker noise coefficient | | 0 | |
| 11 | AF | Flicker noise exponent | | 1 | |
| 12 | FC | Forward-bias depletion cap. coeff. | | 0.5 | |
| 13 | BV | Reverse breakdown voltage | V | ∞ | 50 |
| 14 | IBV | Reverse breakdown current | A | 1E-10 | |

PARÁMETROS DEL MODELO DEL TRANSISTOR MOSFET

| NOMBRE | PARÁMETRO | UNID. | VAL. DEF. | EJEMP. | |
|--------|-----------|--|------------------|----------------------------------|---------|
| 1 | LEVEL | Model type | | 1 | |
| 2 | VTO | Zero-bias threshold voltage | V | 0.0 | 1.0 |
| 3 | KP | Transconductance parameter | A/V ² | 2.0E -5 | 3.1E-5 |
| 4 | GAMMA | Bulk threshold parameter | V ^{0.5} | 0.0 | 0.37 |
| 5 | PHI | Surface potential | V | 0.6 | 0.65 |
| 6 | LAMBDA | Channel-length modulation (LEVEL=1 and 2 only) | V ⁻¹ | 0.0 | 0.02 |
| 7 | RD | Drain ohmic resistance | Ω | 0.0 | 1.0 |
| 8 | RS | Source ohmic resistance | Ω | 0.0 | 1.0 |
| 9 | CBD | Zero-bias B-D junction capacitance | F | 0.0 | 20fF |
| 10 | CBS | Zero-bias B-S junction capacitance | F | 0.0 | 20fF |
| 11 | IS | Bulk junction saturation current | A | 1.0E -14 | 1.0E-15 |
| 12 | PB | Bulk junction potential | V | 0.8 | 0.87 |
| 13 | CGSO | Gate-source overlap capacitance per meter channel width | F/m | 0.0 | 4.0E-11 |
| 14 | CGDO | Gate-drain overlap capacitance per meter channel width | F/m | 0.0 | 4.0E-11 |
| 15 | CGBO | Gate-bulk overlap capacitance per meter channel length | F/m | 0.0 | 2.0E-10 |
| 16 | RSH | Drain and source diffusion sheet resistance | Ω/□ | 0.0 | 10.0 |
| 17 | CJ | zZero-bias bulk junction bottom cap. per sq-meter of junction area | F/m ² | 0.0 | 2.0E-4 |
| 18 | MJ | Bulk junction bottom grading coef. | | 0.5 | 0.5 |
| 19 | CJSW | Zero-bias bulk junction sidewall cap. per meter of junction perimeter | F/m | 0.0 | 1.0E-9 |
| 20 | MJSW | Bulk junction sidewall grading coef. | | 0.50 (level1) 0.33 (level2,3) | |

SPICE: Dispositivos Semiconductores

| NOMBRE | PARÁMETRO | UNID. | VAL. DEF. | EJEMP. |
|----------|---|-----------------------|-----------|---------|
| 21 JS | Bulk junction saturation current per sq-meter of junction area | A/m ² | 1.0E-8 | |
| 22 TOX | Oxide thickness | m | 1.0E -7 | 1.0E-7 |
| 23 NSUB | Substrate doping | cm ⁻³ | 0.0 | 4.0E15 |
| 24 NSS | Surface state density | cm ⁻² | 0.0 | 1.0E10 |
| 25 NFS | Fast surface state density | cm ⁻² | 0.0 | 1.0E10 |
| 26 TPG | Type of gate material: +1 opp. to substrate -1 same as substrate 0 Al gate | | 1.0 | |
| 27 XJ | Metallurgical junction depth | m | 0.0 | 1U |
| 28 LD | Lateral diffusion | m | 0.0 | 0.8U |
| 29 UO | Surface mobility | cm ² /(Vs) | 600 | 700 |
| 30 UCRIT | Critical field for mobility degradation (LEVEL= 2 only) | V/cm | 1.0E 4 | 1.0E4 |
| 31 UEXP | Critical field exponent in mobility degradation (LEVEL= 2 only) | | 0.0 | 0.1 |
| 32 UTRA | Transverse field coef (mobility) (deleted for LEVEL= 2) | | 0.0 | 0.3 |
| 33 VMAX | Maximum drift velocity of carriers | m/s | 0.0 | 5.0E4 |
| 34 NEFF | Total channel charge (fixed and mobile) coefficient (LEVEL= 2 only) | | 1.0 | 5.0 |
| 35 KF | Flicker noise coefficient | | 0.0 | 1.0E-26 |
| 36 AF | Flicker noise exponent | | 1.0 | 1.2 |
| 37 FC | Coefficient for forward-bias depletion capacitance formula | | 0.5 | |
| 38 DELTA | Width effect on threshold voltage (LEVEL= 2 and 3) | | 0.0 | 1.0 |
| 39 THETA | Mobility modulation (LEVEL= 3 only) | V ⁻¹ | 0.0 | 0.1 |
| 40 ETA | Static feedback (LEVEL= 3 only) | | 0.0 | 1.0 |
| 41 KAPPA | Saturation field factor (LEVEL= 3 only) | | 0.2 | 0.5 |

SUBCIRCUITOS

Son circuitos definidos como un conjunto de elementos en el fichero de entrada y que pueden ser llamados y colocados como un dispositivo más dentro de un circuito más complejo.

■ FORMA DE DEFINIRLOS:



```
.SUBCKT  NOMBRE  N1  N2  ...  
(conjunto de elementos y/o otros subcircuitos)  
.ENDS  <NOMBRE>
```

■ FORMA DE INVOCARLOS:

```
XZZZZZZZ  N1  N2  ...  NOMBRE
```

- ⇒ No hay límite a la complejidad o tamaño de un subcircuito.
- ⇒ Cada vez que se invoca un subcircuito, SPICE inserta en el lugar de las líneas de llamada las líneas correspondientes a dicho subcircuito.
- ⇒ Dentro de la definición del subcircuito pueden existir otros subcircuitos y estos a su vez pueden contener otros.
- ⇒ Los nudos definidos en un subcircuito tienen carácter local, salvo el nudo 0 que es global.
- ⇒ Dentro de la definición del subcircuito no se pueden usar líneas de control y sí de modelo.

EJEMPLO



CIRCUITO COMBINACIONAL

```

*
* Descripción del circuito
*
X1 1 2 4 NAND
X2 2 3 5 NAND
X3 4 5 7 NAND
VA 1 0 DC 5
VB 2 0 DC 0
VC 3 0 DC 4.5
*
* Fin de la descripción del circuito
*

```

* Subcircuito NAND

```

*
.SUBCKT NAND 1 2 4
D1 3 1 MOD1
D2 3 2 MOD1
R1 5 3 5K
R2 3 0 5K
Ro 5 4 5K
G1 4 0 POLY 3 0 1 1 1 1
VCC 5 0 DC 5
.MOD MOD1 D
.ENDS NAND
*
.END

```

LÍNEAS DE CONTROL

Permiten especificar el análisis a realizar sobre el circuito, indicando así mismo cómo cambiar parámetros de carácter general.

```
.NOMBRE ....
```

Distinguiremos varios tipos de tarjetas de control:

- Asociadas al análisis en DC.
- Asociadas al análisis en AC.
- Asociadas al análisis TRANSITORIO.
- No asociadas a ningún análisis en particular.

ASOCIADAS AL ANÁLISIS EN DC

.OP

- ⇒ Además de calcular el punto de trabajo (por defecto), da información de los modelos en pequeña señal de los dispositivos.

.DC

```
.DC NOMBRE1 VALC1 VALF1 VALI1  
+ <NOMBRE2 VALC2 VALF2 VALI2>
```

- ⇒ NOMBRE1 y NOMBRE2 son fuentes independientes.
- ⇒ Mediante esta línea se ordena que se haga un análisis en DC para cada valor de las fuentes NOMBRE comprendido entre VALC y VALF en incrementos VALI.
- ⇒ Si se especifican dos fuentes, la primera es barrida sobre su rango para cada valor de la segunda.

ASOCIADAS AL ANÁLISIS EN DC

.NODESET

```
.NODESET V(NUMNOD)=VAL V(NUMNOD)=VAL ....
```

- ⇒ Establece condiciones iniciales en los nudos para que SPICE los use en el análisis en DC.
- ⇒ Puede ser imprescindible para asegurar la convergencia de sistemas o biestables.

.TF

```
.TF VARSAL VALENT
```

- ⇒ Ordena calcular, alrededor del punto de trabajo:
 - la relación en pequeña señal VARSAL/VALENT
 - la impedancia de entrada
 - la impedancia de salida
- ⇒ Como paso previo SPICE hace un análisis en DC.

.SENS

```
.SENS OV1 <OV2 ...>
```

- ⇒ OV1, OV2, ..., son variables de salida.
- ⇒ SPICE calcula la sensibilidad en pequeña señal para cada variable indicada respecto a cada parámetro del circuito.

ASOCIADAS AL ANÁLISIS EN AC

.AC

| | | | | |
|------------------|------------------|-----------------|-------------------|-------------------|
| <code>.AC</code> | <code>DEC</code> | <code>NO</code> | <code>FCOM</code> | <code>FFIN</code> |
| <code>.AC</code> | <code>OCT</code> | <code>NO</code> | <code>FCOM</code> | <code>FFIN</code> |
| <code>.AC</code> | <code>LIN</code> | <code>NO</code> | <code>FCOM</code> | <code>FFIN</code> |


- ⇒ Ordenan un análisis en AC desde una frecuencia FCOM hasta FFIN con la siguiente variación:
 - DEC: Se toman NO frecuencias por década.
 - OCT: Se toman NO frecuencias por octava.
 - LIN: Variación lineal, siendo NO el número total de frecuencias tomadas.

.NOISE

| | | | |
|---------------------|-------------------|--------------------|-------------------|
| <code>.NOISE</code> | <code>OUTV</code> | <code>INSRC</code> | <code>NUMS</code> |
|---------------------|-------------------|--------------------|-------------------|

- ⇒ Ordena el análisis de ruido en el circuito.
- ⇒ OUTV = Tensión de salida que define el punto de suma.
- ⇒ INSRC = Nombre de una fuente independiente que es la referencia de entrada del ruido.
- ⇒ NUMS = Indica el intervalo entre resultados.
- ⇒ Se calcula el ruido equivalente en la entrada y la salida indicadas. Se escribe la contribución de cada generador de ruido a intervalos de frecuencia indicados por NUMS.

EJEMPLO



PAR DIFERENCIAL SIMPLE

*

* Alimentaciones y entradas

*

VCC 7 0 12

VEE 8 0 -12

VIN 1 0 AC 1

*

* Descipcion del circuito

*

RS1 1 2 1K

RS2 6 0 1K

Q1 3 2 4 MOD1

Q2 5 6 4 MOD1

RC1 7 3 10K

RC2 7 5 10K

RE 4 8 10K

*

* Modelos y tarjetas de control

*

.MODEL MOD1 NPN BF=50 VAF=50

+ IS=1.E-12 RB=100 CJC=.5PF TF=.6NS

.AC DEC 10 1 100MEG

.PRINT AC VM(5) VP(5)

.END

*****10/21/93 ***** SPICE 2G.6 3/16/83 *****11:54:01*****

PAR DIFERENCIAL SIMPLE

**** INPUT LISTING TEMPERATURE = 27.000 DEG C

*

* ALIMENTACIONES Y ENTRADAS

*

VCC 7 0 12

VEE 8 0 -12

VIN 1 0 AC 1

*

* DESCIPCION DEL CIRCUITO

*

RS1 1 2 1K

RS2 6 0 1K

Q1 3 2 4 MOD1

Q2 5 6 4 MOD1

RC1 7 3 10K

RC2 7 5 10K

RE 4 8 10K

*

* MODELOS Y TARJETAS DE CONTROL

*

.MODEL MOD1 NPN BF=50 VAF=50 IS=1.E-12 RB=100 CJC=.5PF TF=.6NS

.AC DEC 10 1 100MEG

.PRINT AC VM(5) VP(5)

.END

*****10/21/93 ***** SPICE 2G.6 3/16/83 *****11:54:01*****

PAR DIFERENCIAL SIMPLE

**** BJT MODEL PARAMETERS TEMPERATURE = 27.000 DEG C

MOD1

TYPE NPN

IS 1.00E-12

BF 50.000

NF 1.000

VAF 5.00E+01

BR 1.000

NR 1.000

RB 100.000

TF 6.00E-10

CJC 5.00E-13

*****10/21/93 ***** SPICE 2G.6 3/16/83 *****11:54:01*****

PAR DIFERENCIAL SIMPLE

**** SMALL SIGNAL BIAS SOLUTION TEMPERATURE = 27.000 DEG C

```

NODE VOLTAGE NODE VOLTAGE NODE VOLTAGE NODE VOLTAGE
( 1)  0.0000 ( 2) -0.0100 ( 3)  6.3645 ( 4) -0.5290
NODE VOLTAGE NODE VOLTAGE NODE VOLTAGE NODE VOLTAGE
( 5)  6.3645 ( 6) -0.0100 ( 7) 12.0000 ( 8) -12.0000
    
```

VOLTAGE SOURCE CURRENTS

```

NAME    CURRENT
VCC     -1.127E-03
VEE     1.147E-03
VIN     -9.996E-06
    
```

TOTAL POWER DISSIPATION 2.73E-02 WATTS

*****10/21/93 ***** SPICE 2G.6 3/16/83 *****11:54:01*****

PAR DIFERENCIAL SIMPLE

**** OPERATING POINT INFORMATION TEMPERATURE=27.000DEG C

**** BIPOLAR JUNCTION TRANSISTORS

```

      Q1      Q2
MODEL  MOD1  MOD1
IB     1.00E-05 1.00E-05
IC     5.64E-04 5.64E-04
VBE    0.519  0.519
VBC    -6.374 -6.374
VCE    6.894  6.894
BETADC 56.376 56.376
GM     2.18E-02 2.18E-02
RPI    2.59E+03 2.59E+03
RX     1.00E+02 1.00E+02
RO     1.00E+05 1.00E+05
CPI    1.31E-11 1.31E-11
CMU    2.38E-13 2.38E-13
CBX    0.00E+00 0.00E+00
CCS    0.00E+00 0.00E+00
BETAAC 56.350 56.350
FT     2.60E+08 2.60E+08
    
```

*****10/21/93 ***** SPICE 2G.6 3/16/83 *****11:54:01*****

PAR DIFERENCIAL SIMPLE

**** AC ANALYSIS TEMPERATURE = 27.000 DEG C

| FREQ | VM(5) | VP(5) | FREQ | VM(5) | VP(5) |
|-----------|-----------|------------|-----------|-----------|------------|
| 1.000E+00 | 6.924E+01 | -1.729E-05 | 3.981E+00 | 6.924E+01 | -6.883E-05 |
| 1.259E+00 | 6.924E+01 | -2.177E-05 | 5.012E+00 | 6.924E+01 | -8.665E-05 |
| 1.585E+00 | 6.924E+01 | -2.740E-05 | 6.310E+00 | 6.924E+01 | -1.091E-04 |
| 1.995E+00 | 6.924E+01 | -3.450E-05 | 7.943E+00 | 6.924E+01 | -1.373E-04 |
| 2.512E+00 | 6.924E+01 | -4.343E-05 | 1.000E+01 | 6.924E+01 | -1.729E-04 |
| 3.162E+00 | 6.924E+01 | -5.467E-05 | 1.259E+01 | 6.924E+01 | -2.177E-04 |


SPICE: Líneas de Control

| FREQ | VM(5) | VP(5) | FREQ | VM(5) | VP(5) |
|-----------|-----------|------------|-----------|-----------|------------|
| 1.585E+01 | 6.924E+01 | -2.740E-04 | 5.012E+05 | 6.848E+01 | -8.602E+00 |
| 1.995E+01 | 6.924E+01 | -3.450E-04 | 6.310E+05 | 6.804E+01 | -1.078E+01 |
| 2.512E+01 | 6.924E+01 | -4.343E-04 | 7.943E+05 | 6.737E+01 | -1.349E+01 |
| 3.162E+01 | 6.924E+01 | -5.467E-04 | 1.000E+06 | 6.635E+01 | -1.681E+01 |
| 3.981E+01 | 6.924E+01 | -6.883E-04 | 1.259E+06 | 6.482E+01 | -2.084E+01 |
| 5.012E+01 | 6.924E+01 | -8.665E-04 | 1.585E+06 | 6.260E+01 | -2.562E+01 |
| 6.310E+01 | 6.924E+01 | -1.091E-03 | 1.995E+06 | 5.950E+01 | -3.117E+01 |
| 7.943E+01 | 6.924E+01 | -1.373E-03 | 2.512E+06 | 5.542E+01 | -3.736E+01 |
| 1.000E+02 | 6.924E+01 | -1.729E-03 | 3.162E+06 | 5.037E+01 | -4.399E+01 |
| 1.259E+02 | 6.924E+01 | -2.177E-03 | 3.981E+06 | 4.459E+01 | -5.074E+01 |
| 1.585E+02 | 6.924E+01 | -2.740E-03 | 5.012E+06 | 3.848E+01 | -5.728E+01 |
| 1.995E+02 | 6.924E+01 | -3.450E-03 | 6.310E+06 | 3.245E+01 | -6.336E+01 |
| 2.512E+02 | 6.924E+01 | -4.343E-03 | 7.943E+06 | 2.686E+01 | -6.882E+01 |
| 3.162E+02 | 6.924E+01 | -5.467E-03 | 1.000E+07 | 2.192E+01 | -7.360E+01 |
| 3.981E+02 | 6.924E+01 | -6.883E-03 | 1.259E+07 | 1.770E+01 | -7.776E+01 |
| 5.012E+02 | 6.924E+01 | -8.665E-03 | 1.585E+07 | 1.417E+01 | -8.137E+01 |
| 6.310E+02 | 6.924E+01 | -1.091E-02 | 1.995E+07 | 1.127E+01 | -8.454E+01 |
| 7.943E+02 | 6.924E+01 | -1.373E-02 | 2.512E+07 | 8.908E+00 | -8.734E+01 |
| 1.000E+03 | 6.924E+01 | -1.729E-02 | 3.162E+07 | 7.000E+00 | -8.985E+01 |
| 1.259E+03 | 6.924E+01 | -2.177E-02 | 3.981E+07 | 5.465E+00 | -9.207E+01 |
| 1.585E+03 | 6.924E+01 | -2.740E-02 | 5.012E+07 | 4.239E+00 | -9.400E+01 |
| 1.995E+03 | 6.924E+01 | -3.450E-02 | 6.310E+07 | 3.265E+00 | -9.563E+01 |
| 2.512E+03 | 6.924E+01 | -4.343E-02 | 7.943E+07 | 2.497E+00 | -9.697E+01 |
| 3.162E+03 | 6.924E+01 | -5.467E-02 | 1.000E+08 | 1.890E+00 | -9.801E+01 |
| 3.981E+03 | 6.924E+01 | -6.883E-02 | | | |
| 5.012E+03 | 6.924E+01 | -8.665E-02 | | | |
| 6.310E+03 | 6.924E+01 | -1.091E-01 | | | |
| 7.943E+03 | 6.924E+01 | -1.373E-01 | | | |
| 1.000E+04 | 6.924E+01 | -1.729E-01 | | | |
| 1.259E+04 | 6.924E+01 | -2.177E-01 | | | |
| 1.585E+04 | 6.924E+01 | -2.740E-01 | | | |
| 1.995E+04 | 6.924E+01 | -3.450E-01 | | | |
| 2.512E+04 | 6.924E+01 | -4.343E-01 | | | |
| 3.162E+04 | 6.923E+01 | -5.467E-01 | | | |
| 3.981E+04 | 6.923E+01 | -6.883E-01 | | | |
| 5.012E+04 | 6.923E+01 | -8.665E-01 | | | |
| 6.310E+04 | 6.923E+01 | -1.091E+00 | | | |
| 7.943E+04 | 6.922E+01 | -1.373E+00 | | | |
| 1.000E+05 | 6.921E+01 | -1.728E+00 | | | |
| 1.259E+05 | 6.919E+01 | -2.176E+00 | | | |
| 1.585E+05 | 6.916E+01 | -2.738E+00 | | | |
| 1.995E+05 | 6.912E+01 | -3.446E+00 | | | |
| 2.512E+05 | 6.904E+01 | -4.335E+00 | | | |
| 3.162E+05 | 6.893E+01 | -5.451E+00 | | | |
| 3.981E+05 | 6.876E+01 | -6.851E+00 | | | |


JOB CONCLUDED

TOTAL JOB TIME 0.23

VM(5)



VP(5)



ASOCIADAS AL ANÁLISIS EN AC

.DISTO

.DISTO RLOAD <INTER <SKW2 <REFPWR <SPW2 >

- ⇒ Ordena un análisis de distorsión como parte del análisis en AC.
- ⇒ Se consideran dos frecuencias distintas:
 - la nominal F1 del análisis AC
 - $F2 = SKW2 \cdot F1$
- ⇒ Se pueden obtener las siguientes medidas de distorsión:
 - HD2 → magnitud a $2 \cdot F1$ con F2 ausente
 - HD3 → magnitud a $3 \cdot F1$ con F2 ausente
 - SIM2 → magnitud a $F1 + F2$
 - DIM2 → magnitud a $F1 - F2$
 - DIM3 → magnitud a $2 \cdot F1 - F2$
- ⇒ RLOAD = Nombre de una resistencia de carga en la que se calculan los productos de potencia de distorsión.
- ⇒ INTER = Intervalo de frecuencias donde se da el resumen de la contribución de todos los dispositivos no-lineales a la distorsión total.
- ⇒ REFPWR = Nivel de potencia de referencia usado en el cálculo de productos de distorsión.
- ⇒ SPW2 = Amplitud de F2

Valores por defecto:

INTER=no se dan resultados
SKW2=0.9
REFPWR=1MW
SPW2=1

ASOCIADAS AL ANÁLISIS TRANSITORIO

.IC

```
.IC V(NUMNOD)=VAL V(NUMNOD)=VAL ....
```

- ⇒ Especifica condiciones iniciales para el análisis transitorio. Éstas se usan si se especifica la opción UIC; en caso contrario las condiciones iniciales se establecen mediante un análisis en DC.

.TRAN

```
.TRAN TSTEP TSTOP <TSTART <TMAX <UIC >
```

- ⇒ Especifica las condiciones del análisis transitorio:
 - TSTEP → Incremento temporal para imprimir resultados
 - TSTOP → Tiempo final
 - TSTART → Tiempo inicial (para imprimir) (por defecto = 0)
 - TMAX → Máximo incremento temporal para análisis interno. Define el mínimo TSTEP. (por defecto es (TSTOP-TSTART)/50)
 - UIC → Indica que se usen las condiciones iniciales especificadas en .IC

.FOUR

```
.FOUR FREQ OV1 <OV2 OV3 ...>
```

- ⇒ Se pide un análisis de Fourier como parte del análisis transitorio.
- ⇒ OV1, OV2, OV3, ..., son nombres de variables de salida.
- ⇒ El análisis se realiza sobre datos del intervalo:

$$TSTOP - \text{Periodo} < t < TSTOP, \quad \text{Periodo} = 1/\text{FREQ}$$
- ⇒ Los resultados suministrados son la componente en DC y los 9 primeros armónicos.

LÍNEAS DE CONTROL DE SALIDA

SPICE puede suministrar dos tipos de salida:

Listado → .PRINT

Dibujo → .PLOT

```
.PRINT TIPO OV1 <OV2 OV3 ... OV8>
.PLOT TIPO OV1<(L1,H1)> <OV2<(L2,H2)> ... OV8<(L8,H8)>>
```

- ⇒ TIPO: Tipo de análisis (AC, DC, TRAN, NOISE, DISTO).
- ⇒ OV1, ..., son los nombres de variables de salida. Pueden ser:
 - Tensiones en nudos o entre dos nudos. (Ej.: V(5), V(7,3))
 - Intensidades en fuentes independientes de tensión. (Ej.: I(Vcc), I(Vin))
 - Para el análisis en AC:
 - VR o IR → Parte real
 - VI o II → Parte imaginaria
 - VM o IM → Módulo
 - VP o IP → Fase
 - VDB o IDB → Decibelios ($20 \cdot \log_{10}(\text{Módulo})$)

Ejemplos: VR(5), IM(Vin), VDB(7)

- ⇒ (Li,Hi) representan la variable OVi dentro de esa escala. Si no se indica se supone autoescalado.

Ejemplos: .PLOT TRAN V(17,5) I(Vin) V(17)

OTRAS LÍNEAS DE CONTROL

.TEMP

```
.TEMP T1 <T2 <T3 ... >
```

- ⇒ Especifica las temperaturas a las que el circuito será simulado. T1, T2, T3, ..., son temperaturas en grados centígrados (°C).
- ⇒ Si se omite entonces la temperatura de simulación es la especificada con TNOM en la línea de control .OPTIONS. (Si ésta a su vez se omite, la temperatura por defecto es de 27°C)

.WIDTH

```
.WIDTH IN=NCOLUM OUT=NCOLUM
```

- ⇒ Define el número máximo de columnas de los ficheros de entrada y salida.

.OPTIONS

```
.OPTIONS OPT1 OPT2 ... (o OPT=OPTVAL ...)
```

- ⇒ Esta línea permite al usuario cambiar determinadas opciones y permite ajustar el control del programa.
- ⇒ Hay 32 opciones diferentes:
 - Algunas opciones se usan para controlar la salida. (Ej.: NOPAGE, LIMPTS=x)
 - Otras opciones se usan para controlar el tipo de análisis que realiza SPICE.
 - Otras para obtener convergencia, etc.


OPCIONES

EFEECTO

| | |
|----------|--|
| GMIN=x | Resets the value of GMIN, the minimum conductance allowed by the program. The default value is 1.0E-12. |
| RELTOL=x | Resets the relative error tolerance of the program. The default value is 0.001 (0.1%). |
| ABSTOL=x | Resets the absolute current error tolerance of the program. The default value is 1pA. |
| VNTOL=x | Resets the absolute voltage error tolerance of the program. The default value is 1μV. |
| TRTOL=x | Resets the transient error tolerance. The default value is 7.0. This parameter is an estimate of the factor by which SPICE overestimates the actual truncation error. |
| CHGTOL=x | Resets the charge tolerance of the program. The default value is 1.0E-14. |
| PIVTOL=x | Resets the absolute minimum value for a matrix entry to be accepted as a pivot. The default value is 1.0E-13. |
| PIVREL=x | Resets the relative ratio between the largest column entry and an acceptable pivot value. The default value is 1.0E-3. In the numerical pivoting algorithm the allowed minimum pivot value is determined by $\text{EPSREL}=\text{AMAX1}(\text{PIVREL}*\text{MAXVAL},\text{PIVTOL})$ where MAXVAL is the maximum element in the column where a pivot is sought (partial pivoting). |
| TNOM=x | Resets the nominal temperature. The default value is 27°C (300 °K). |
| ITL1=x | Resets the dc iteration limit. The default is 100. |
| ITL2=x | Resets the dc transfer curve iteration limit. The default is 50. |
| ITL5=x | Resets the transient analysis total iteration limit the default is 5000. Set ITL5=0 to omit this test. |
| DEFL=x | Resets the value for MOS channel length; the default is 100.0μm. |
| DEFW=x | Resets the value for MOS channel width; the default is 100.0μm. |
| DEFAD=x | Resets the value for MOS drain diffusion area; the default is 0.0. |
| DEFAS=x | Resets the value for MOS source diffusion area; the default is 0.0. |

ANÁLISIS EN DC: .DC



inversor cmos - tipos de analisis
100/03/0 12:39:37



ANÁLISIS EN DC: .OP

**** mosfets

| element | mn | mp |
|---------|------------|------------|
| model | nmos | pmos |
| id | 311.5309u | -311.5310u |
| ibs | 0 | 0 |
| ibd | -101.8000p | 2.272e-17 |
| vgs | 1.6000 | -1.7000 |
| vds | 1.7502 | -1.5498 |
| vbs | 0 | 0 |
| vth | 573.9130m | -550.4180m |
| vdsat | 830.3756m | -981.8583m |
| beta | 115.0323u | 157.8427u |
| gam eff | 527.6252m | 527.6252m |
| gm | 440.4485u | 437.4501u |
| gds | 14.5032u | 29.0124u |
| gmb | 76.2947u | 120.8119u |
| cdtot | 2.6143f | 6.5795f |
| cgdot | 2.5567f | 5.5348f |
| cstot | 5.0148f | 13.2306f |
| cbtot | 6.2974f | 16.9915f |
| cgs | 2.3439f | 4.9162f |
| cgd | 1.455e-16 | 4.123e-16 |



ANÁLISIS EN DC: .TF

**** small-signal transfer characteristics


$$v(\text{out})/v_{in} = -20.1737$$

$$\text{input resistance at } v_{in} = 1.000e+20$$


$$\text{output resistance at } v(\text{out}) = 22.9795k$$

ANÁLISIS TRANSITORIO: .TRAN

inversor cmos - tipos de analisis
100/03/0 12:39:37



ANÁLISIS EN AC: .AC



ANÁLISIS EN DC: .TF
 $V_{out}/V_{in} = -20.17 = 26.09dB$

ANÁLISIS EN AC: .NOISE


**** Results of the sqrt of integral (v**2 / freq)

**** total output noise voltage = 1.7394m volts

**** total equivalent input noise = 120.0657u


ANÁLISIS TRANSITORIO: .TRAN

Relación con el Diagrama de Bode (1)




ANÁLISIS TRANSITORIO: .TRAN

Relación con el Diagrama de Bode (2)




ANÁLISIS TRANSITORIO: .TRAN

Relación con el Diagrama de Bode (3)



ANÁLISIS EN DC: .DC

inversor cmos - tipos de analisis
100/03/0 12:39:37



ANÁLISIS TRANSITORIO: .FOUR

***** fourier analysis tnom= 25.000 temp= 25.000


fourier components of transient response v(out)
dc component = 1.667E+00

| no | frequency (hz) | fourier component | normalized phase component (deg) | normalized phase (deg) |
|----|----------------|-------------------|----------------------------------|------------------------|
|----|----------------|-------------------|----------------------------------|------------------------|

| | | | | | |
|---|---------|-----------|-----------|------------|-----------|
| 1 | 1.0000k | 855.7490m | 1.0000 | 179.9985 | 0. |
| 2 | 2.0000k | 68.8141m | 80.4139m | 89.9990 | -89.9995 |
| 3 | 3.0000k | 55.5648m | 64.9312m | -179.9929 | -359.9913 |
| 4 | 4.0000k | 15.3072m | 17.8875m | 90.0207 | -89.9778 |
| 5 | 5.0000k | 1.3988m | 1.6345m | -438.7839m | -180.4372 |
| 6 | 6.0000k | 374.9768u | 438.1855u | -90.9938 | -270.9922 |
| 7 | 7.0000k | 1.3253m | 1.5487m | -158.0425m | -180.1565 |
| 8 | 8.0000k | 955.4913u | 1.1166m | -90.1097 | -270.1082 |
| 9 | 9.0000k | 190.5515u | 222.6722u | -179.7079 | -359.7063 |


total harmonic distortion = 10.4924 percent

ANÁLISIS TRANSITORIO: .TRAN



ANÁLISIS EN DC: .DC

inversor cmos - tipos de analisis
100/03/0 12:39:37



ANÁLISIS TRANSITORIO: .FOUR

***** fourier analysis tnom= 25.000 temp= 25.000


fourier components of transient response v(out)
dc component = 1.618E+00

| no | frequency (hz) | fourier component | normalized phase (deg) | normalized phase (deg) |
|----|----------------|-------------------|------------------------|------------------------|
|----|----------------|-------------------|------------------------|------------------------|

| | | | | | |
|---|---------|-----------|-----------|-----------|-----------|
| 1 | 1.0000k | 1.5191 | 1.0000 | 179.9949 | 0. |
| 2 | 2.0000k | 15.6135m | 10.2784m | 89.9528 | -90.0421 |
| 3 | 3.0000k | 338.4282m | 222.7886m | 179.9847 | -10.2386m |
| 4 | 4.0000k | 22.2844m | 14.6699m | 89.9400 | -90.0549 |
| 5 | 5.0000k | 157.1142m | 103.4289m | 179.9838 | -11.0779m |
| 6 | 6.0000k | 20.8522m | 13.7271m | 89.9233 | -90.0716 |
| 7 | 7.0000k | 87.1199m | 57.3514m | 179.9983 | 3.4485m |
| 8 | 8.0000k | 18.1878m | 11.9731m | 89.9160 | -90.0789 |
| 9 | 9.0000k | 51.3885m | 33.8293m | -179.9655 | -359.9604 |

total harmonic distortion = 25.5771 percent

ANÁLISIS TRANSITORIO: .TRAN





Ejemplo: Circuito RC

Circuito RC

```
vin in 0 ac pulse (0 5 100n 5n 5n 1u 2u)
res in out 10k
cap out 0 10p
```

```
.tran 5n 4u
.ac dec 10 1 1g
.end
```




Ejemplo: Circuito Rectificador

Circuito Rectificador



```

vin in 0 sin(0 12 50 0 0)
d1 in out1 dm
r1 out1 0 10k



d2 in out2 dm
r2 out2 0 10k
cap out2 0 10u
.model dm d cjo=2p rs=1 is=1p
.tran 0.1m 0.06
.end
    
```



1) NUDOS CON MENOS DE 2 CONEXIONES







2) NUDOS FLOTANTES → Sin camino de DC a tierra



3) LAZOS DE FUENTES Y/O INDUCTORES


Lazos de resistencia cero: V, L, VCVS, CCVS



4) VALORES NEGATIVOS DE COMPONENTES

Se permiten valores < 0 de R, C y L → .OP, .DC, .AC ✓

.TRAN ~ Posible inestabilidad



```
**warning** 0:r1 resistance is negative. it may cause instability problem.
**warning** 0:r2 resistance is negative. it may cause instability problem.
```

```
node =voltage node =voltage
+0:1 = 1.0000 0:2 = 714.2857m
```

```
**** small-signal transfer characteristics
v(2)/vin = 714.2857m
input resistance at vin = -140.0000
output resistance at v(2) = -28.5714
```